A primeira função a ser mostrada é a que fará o cálculos dos mínimos quadrados. Esta função foi passada em aula pela professora e permaneceu inalterada para o trabalho.

quadrados_minimos.sci | usando_Met_Qua_Min.sce

```
function [a] = quadrados_minimos(X,F,GLista)

n=size(GLista);
for i=1:n
    gi= GLista(i)(X);
    for j=i:n
        gj=GLista(j)(X);
        G(i,j)=gi*gj';
        G(j,i)=G(i,j);
    end
    b(i)=F*gi';
end
a=G\b
endfunction

```

Em seguida, há o script que usará a função mostrada. Nele, é necessário primeiro definir g1, g2 e g3, que serão cruciais para definir os gráficos de reta e parábola. Também é necessário definir os valores de acordo com a tabela passada.

quadrados_minimos.sci | usando_Met_Qua_Min.sce

```
// usando a função quadrados_minimos para encontrar
// uma parábola que aproximar os valores
// x = [0, 0.25, 0.5, 0.75, 1, 1.25, 1.5, 1.75, 2]
// f(x) = [-1.8, -1.2, -0.4, 0.4, 1.1, 2.1, 3, 3.9, 5]
// e também plotar o gráfico dos pontos de da G(x)

function [z]=g1(X)
    z = X.^0
endfunction

function [z]=g2(X)
    z = X
endfunction

function [z]=g3(X)
    z = X.^2
endfunction

exec('quadrados_minimos.sci');
//definindo os pontos tabelados da função
X = [0, 0.25, 0.5, 0.75, 1, 1.25, 1.5, 1.75, 2]
F = [-1.8, -1.2, -0.4, 0.4, 1.1, 2.1, 3, 3.9, 5]

```

Após serem definidos os valores e funções necessárias para os cálculos, a montagem da lista que será usada pela função é feita, possibilitando o cálculo da reta, da parábola e seus erros.

quadrados_minimos.sci | usando_Met_Qua_Min.sce

```
HLista = list(g1,g2)
[b] = quadrados_minimos(X,F,HLista)
mprintf('Reta:')
disp(b)

//Cálculo do erro da reta
HX = b(1) + b(2)*X;
Y = F-HX;
E = Y*Y'
mprintf('Erro da reta:')
disp(E)

GLista=list(g1,g2,g3)
[a] = quadrados_minimos(X,F,GLista)
mprintf('Parábola:')
disp(a)

//Cálculo do erro da parábola
GX = a(1) + a(2)*X + a(3)*X.^2;
Y = F-GX;
E = Y*Y'
mprintf('Erro da parábola:')
disp(E)
```

Os cálculos são feitos, sendo então mostrados no console. É possível perceber que a parábola tem um erro menor que a reta.

```
--> exec('C:\Users\ademi\Desktop\Trab3\usando_Met_Qua_Min.sce', -1)
Reta:
  -2.0555556
   3.4
Erro da reta:
   0.2122222
Parábola:
  -1.8212121
   2.5965368
   0.4017316
Erro da parábola:
   0.0180519
```

Após isso, o script gera os gráficos com os pontos, reta e parábola encontrados.

```

// gráficos
x=linspace(-0.5,2.5,100);

// Reta encontrada
H = b(1) + b(2)*x;
plot(x, H, "blue");

// Parábola encontrada
G = a(1) + a(2)*x + a(3)*x.^2;
plot(x,G, "green");

// Pontos encontrados
plot(X,F,'ro');
title("Reta e Parabola");

```

Os dados são mostrados em diferentes cores para facilitar a visualização, sendo a reta em cor azul, a parábola em cor verde e os pontos em vermelho.



Para finalizar, é possível analisar qual é a melhor aproximação. Considerando que o erro da parábola é menor e o objetivo seja conseguir um gráfico mais próximo dos pontos, a parábola seria a aproximação indicada. Um dos possíveis motivos para isso seria que a parábola usa e calcula mais dados para fazer sua aproximação. O custo computacional maior para o cálculo da parábola se mostra pouco relevante em comparação com suas vantagens.